
// PROJECT  :LCDTMP36
// PURPOSE  :Monitor the readings from a TMP36 on an LCD Screen
// COURSE   :ICS3U
// AUTHOR   :C. D'Arcy
// DATE     :2025 01 28
// MCU      :328P
// STATUS   :Working
// REFERENCE:LCD: https://www.arduino.cc/en/Reference/LiquidCrystal
//          :LCD: https://www.arduino.cc/en/Tutorial/HelloWorld
//          :HD44780U: http://darcy.rsgc.on.ca/ACES/Datasheets/HD44780.pdf
//          :TMP36: http://darcy.rsgc.on.ca/ACES/Datasheets/TMP35_36_37.pdf

#include <LiquidCrystal.h>  //Adafruit's LCD Library...

[bookmark: _GoBack]//Pin assignments below are RPN Calculator Backplane-compatible
uint8_t RS = 9, EN = 8, D4 = 7, D5 = 6, D6 = 5, D7 = 4;
LiquidCrystal lcd(RS, EN, D4, D5, D6, D7);

#define LCD_COLUMNS 16      //number of columns in Character LCD screen
#define LCD_ROWS 2          //number of rows on LCD screen
#define DEGREE 0xDF         //Code for degree symbol (see HD44780U Datasheet) 
#define TMP36 A5            //ADC pin for sensor 
#define VREF 5.0            //improved accuracy for this particular sensor (TMP36)
 
void setup() {
  lcd.begin(LCD_COLUMNS, LCD_ROWS);   //initialize the LCD screen
  lcd.clear();                        //erase the LCD screen and send cursor home

  lcd.print("TMP36 SENSOR");          //provide a title for the first row
  lcd.setCursor(0, 1);                //position the cursor at start of second row
  lcd.print("C:    ");                //a label
  lcd.write(DEGREE);                  //display the degree symbol
  lcd.setCursor(LCD_COLUMNS>>1, 1);   //position the cursor for another label halfway across
  lcd.print("F:    ");                //display another label
  lcd.write(DEGREE);                  //display the degree symbol
}

void display(float c, float f) {      //great code is modular...
  char buffer[6];                     //storage for float converted to a string
  lcd.setCursor(2, 1);                //prepare to update the data
  dtostrf(c, 4, 1, buffer);           //convert float (celsius) to string
  lcd.print(buffer);                  //display it
  lcd.setCursor(10, 1);               //position cursor
  dtostrf(f, 4, 1, buffer);           //convert float (fahrenheit) to string
  lcd.print(buffer);                  //display it
}

void loop() {
  uint16_t reading = analogRead(TMP36);         //obtain sensor (pot) ADC reading
  float voltage = reading * VREF / 1024;        //obtain voltage value if interested
  float celsius = (voltage * 1000 - 500) / 10;  //convert voltage to celsius equivalent 
  float fahrenheit = celsius * 1.8 + 32;        //determine fahrenheit equivalent
  display(celsius, fahrenheit);                 //display results on the LCD screen
}

